



PAR FRANÇOIS LE GRIGUER

JOYSTICK FICHER

PREMIER COUP DE CHAPEAU A GUY PARRENIN

La nouvelle formule prend un très bon départ sur AMSTRAD, Guy PARRENIN qui précise n'avoir aucune notion d'informatique, vient de nous envoyer le premier listing de «gestion de fichier» sur Amstrad, un programme complet prévu pour deux types de fichiers: les JEUX et les UTILITAIRES. Un logiciel simple à utiliser et bien présenté que l'on découvre ensemble en y ajoutant tous les commentaires en italique pour toi qui débute (ne saisis surtout pas les textes en italique).

**QUI EST LE MEILLEUR?
DEPUIS LA SEMAINE DERNIERE
UNE VERITABLE COMPETITION
S'EST ENGAGEE ENTRE
TOUS LES JOYSTICKERS FOUS DE
PROGRAMMATION AVEC
LA REALISATION
D'UNE GESTION DE FICHER.
QUE TU UTILISES UN AMSTRAD,
UN C64, UN ATARI, UN
AMIGA, UN PC,
UN THOMSON ETC...
TU ES CONCERNE.
TOUTES LES SEMAINES LES
MEILLEURS SOFTS RECUS VONT
SERVIR DE BASE A J'APPRENDS
EN INCORPORANT TOUTES LES
EXPLICATIONS A VOS LISTINGS.
SI TU ES GENIAL TON NOM
VA FIGURER AU TOP ET
POURQUOI PAS TA PHOTO
ENCORE PLUS SUPER NON!!!
AUJOURD'HUI LE
PREMIER COUP DE
CHAPEAU SUR AMSTRAD.**

J'APPRENDS

```

10 ***** LOGITHEQUE *****
20 'copyright JOYSTICK Hebdo et Guy PARRENIN
30 '
40 CLS:MODE 2
50 BORDER 14:INK 0,14:INK 1,0
60 LOCATE 15,12:PRINT CHR$(24);
PRINT CHR$(24) donne un affichage en inversion vidéo.
70 PRINT» CHARGER VOTRE FICHER AVANT D'AJOUTER DES
FICHES. «;
80 PRINT CHR$(24):PRINT CHR$(7)
PRINT CHR$(7) émet le bip sonore.
90 OPENOUT»GUY»:MEMORY HIMEM-1:CLOSEOUT
Cette ligne avec OPENOUT et CLOSEOUT teste la présence de la
disquette.
100 FOR T=0 TO 500:NEXT T
Boucle de tempo (délai d'affichage du message).
110 DIM A$(150,3)
DIMensionner le fichier A$ pour 150 enregistrements contenant
chacun 3 zones. Si ton fichier est plus important tu remplaces 150
par le nombre prévu.
120 POKE &B632,255
Ce poke transforme les saisies minuscules en majuscules. Pour un
CPC 464 mettre POKE &B4E8,255.
130 CLS
140 IV$=CHR$(24):Q$="" OK ? O/N «:T1$=""
Déclaration de chaînes de caractères.
150 DATA 1,CHARGEMENT,2,AJOUT,3,LISTE ECRAN
160 DATA 4,MODIFICATIONS,5,SAUVEGARDE
170 DATA 6,IMPRESSION,7,EFFACEMENT MEMOIRE,8,RETOUR
BASIC
Ces DATA contiennent les textes du menu.
180 RESTORE 150
Pour indiquer au système que la 1ère DATA à lire est en ligne 150.
190 N=8
200 T1$="" MENU LOGITHEQUE «
210 LOCATE(80-LEN(T1$))/2,2:PRINT IV$:T1$:IV$
Une méthode facile pour calculer la valeur en x, c-à-d, centrer le texte
à afficher. 80 représente la largeur totale de l'écran. LEN(T1$) donne
le nombre de caractères de la chaîne T1$. 80 - longueur texte / 2 =
espace à laisser à gauche. L'inversion vidéo est donnée par IV$ (IV$
= chr$(24)).
220 T0$=""
230 FOR J=1 TO N
240 READ O$,O0$
Lecture des DATA par groupe de 2. (numéro et texte ligne menu).
250 T0$=T0$+O$
La chaîne T0$ est mise à jour.
260 LOCATE 30,J*2+12-N:PRINT O$;» «:CHR$(154);CHR$(243);»
«:O0$
Affichage de la ligne menu avec calcul de y en fonction de la valeur
de J.
270 NEXT
280 LT=LEN(T0$)
Définition en LT de la longueur de la chaîne T0$.
290 LOCATE 32-LT,24:PRINT IV$;» VOTRE CHOIX («;
300 FOR I=1 TO LT-1
310 PRINT MID$(T0$,I,1);»»;

```

La fonction MID\$ extrait de la chaîne T0\$ à partir du numéro de caractère indiqué par I, 1 caractère.

```
320 NEXT
330 PRINT RIGHT$(T0$,1);-);CHR$(154);CHR$(243);-
-);IV$;CHR$(207)
```

La fonction RIGHT\$ extrait les caractères de droite de la chaîne T0\$. Ici 1 seul donc le dernier.

```
340 R$=-);WHILE R$=-);R$=INKEY$:WEND
350 K=INSTR(T0$,R$)
360 IF K=0 THEN PRINT CHR$(7);GOTO 340
Attente d'une composition au clavier et bip sonore si erreur. INSTR
recherche si la chaîne T0$ contient R$ (composé). Si la valeur
composée est OK suite dans la séquence par ON K GOTO. Si k=1
GOTO EN 380 etc...
```

```
370 ON K GOTO 380,780,1220,1450,1970,2380,3330,3490
En conclusion toute cette séquence d'affichage du menu repré-
sente un bel exercice de style en mettant en évidence les possibil-
ités du Basic. Elle pouvait s'écrire beaucoup plus simplement. En
programmation l'objectif doit toujours être: le plus court et le plus
simple.
```

```
380 '-----CHARGEMENT-----
390 CLS
400 T1$=-); CHARGEMENT -
410 LOCATE(80-LEN(T1$))/2,6:PRINT IV$;T1$;IV$
420 T2$=-); DE VOTRE LOGITHEQUE -
430 LOCATE(80-LEN(T2$))/2,8:PRINT IV$;T2$;IV$
440 T3$=-); JEUX OU UTILITAIRES ? (J/U) -
450 LOCATE(80-LEN(T3$))/2,13:PRINT T3$
460 LOCATE 14,14:PRINT STRING$(7,154);SPACES$(8);
470 PRINT STRING$(21,154);SPACES$(9);STRING$(5,154)
480 Z$=INKEY$:IF Z$=-); THEN 510
490 IF Z$=-);J THEN 520
500 IF Z$=-);U THEN 650
510 GOTO 480
```

Le chargement ne pose jusqu'ici aucun problème particulier il s'agit simplement d'affichage déjà longuement commenté et, ensuite de la composition d'un J ou U à tester.

```
520 T4$=-); J E U X -
530 LOCATE(80-LEN(T4$))/2,19:PRINT IV$;T4$;IV$
540 OPENIN "JEUX.LOG"
550 INPUT#9,N0
```

Ouverture du fichier JEUX et lecture de la première zone «N0» qui contient le nombre d'enregistrements du fichier, valeur mise à jour lors de la sauvegarde.

```
560 FOR I=1 TO N0
Début de la boucle de programme pour le nombre d'enregistre-
ments.
```

```
570 FOR J=1 TO 3
Début de la boucle de programme pour le nombre de zones à
l'intérieur de chaque enregistrement.
```

```
580 K=N2+1
590 INPUT#9,A$(K,J)
Mise à jour du fichier A$ (déclaré en début de traitement).
```

```
600 NEXT
610 NEXT
620 CLOSEIN
630 N2=N2+N0
640 PRINT CHR$(7);GOTO 130
```

Fermeture du fichier, maj de N2 pour le nombre d'enregistrements, bip sonore et retour au menu.

```
650 T4$=-); UTILITAIRES -
Procédures identiques au chargement du fichier JEUX.
660 LOCATE(80-LEN(T4$))/2,19:PRINT IV$;T4$;IV$
670 OPENIN "UTILIT.LOG"
680 INPUT#9,N0
690 FOR I=1 TO N0
700 FOR J=1 TO 3
710 K=N2+1
720 INPUT#9,A$(K,J)
730 NEXT
740 NEXT
750 CLOSEIN
760 N2=N2+N0
770 PRINT CHR$(7);GOTO 130
780 '-----AJOUT-----
```

Cette séquence permet de créer de nouveaux enregistrements dans un fichier déjà existant ou sa création complète.

```
790 CLS
800 T1$=-); AJOUT LOGITHEQUE -
810 LOCATE(80-LEN(T1$))/2,4:PRINT IV$;T1$;IV$
```

```
820 LOCATE 22,9:PRINT-NOM :-
830 LOCATE 22,11:PRINT-DISC :-
840 LOCATE 22,13:PRINT-OBS. :-
850 WINDOW#1,28,60,9,14
860 CLS#1
```

Ouverture d'une fenêtre avec un numéro de canal (#1), les coordonnées début et fin en X puis en Y. Effacement de la fenêtre (CLS#1). Le gros avantage d'une fenêtre c'est de permettre une gestion indépendante dans toutes les fonctions d'affichage: couleurs, effacement, scrolling...

```
870 T1$=-); DISC : 1 -> 999 ENTER : RETOUR AU MENU -
880 LOCATE(80-LEN(T1$))/2,20:PRINT IV$;T1$;IV$
890 LOCATE 1,23:PRINT CHR$(18)
900 LOCATE 29,9:PRINT"....."
910 LOCATE 29,11:PRINT"...."
920 LOCATE 29,13:PRINT"....."
Affichage à l'écran des 3 lignes de saisie.
930 FOR I=1 TO 3
940 IF I=1 THEN LOCATE 29,9
950 IF I=2 THEN LOCATE 29,11
960 IF I=3 THEN LOCATE 29,13
970 LINE INPUT"-);S$(I)
```

Saisie de la ligne.

```
980 IF S$(1)=-);OR LEFT$(S$(1),1)=-); THEN 130
Retour en menu sans composition de texte.
990 LOCATE 44,20:PRINT IV$;SPACES$(23);IV$
1000 NEXT
1010 LOCATE(80-LEN(Q$))/2,23:PRINT IV$;Q$;IV$
1020 Z$=INKEY$:IF Z$=-); THEN 1020
1030 IF Z$=-);N THEN 860 ELSE IF Z$<>O THEN 1010
Validation par O ou refus par N.
```

```
1040 N2=N2+1
Mise à jour du nombre d'enregistrements.
1050 FOR I=1 TO 3
1060 A$(N2,I)=S$(I)
```

Boucle de mise à jour du fichier A\$ pour les 3 zones de l'enregistrement. Le nouveau titre entre en fin de fichier avant le tri.

```
1070 NEXT
1080 FOR I=1 TO N2
Début de la boucle de tri du fichier. A chaque entrée d'un nouveau titre
il s'agit de le replacer par ordre alphabétique en le comparant à tous
les autres titres déjà en fichier.
```

```
1090 IF A$(I,1)<S$(1) THEN 1200
Si le titre du fichier est < que le nouveau titre pas de traitement.
1100 J=N2
```

```
1110 FOR K=1 TO 3
1120 A$(J,K)=A$(I-1,K)
```

Le contenu du fichier se décale vers le bas. L'enregistrement reçoit le contenu de celui qui le précède, pour les 3 zones, et jusqu'à l'enregistrement en cours de comparaison.

```
1130 NEXT K
1140 J=J-1
1150 IF J>1 THEN 1110
```

Ici tous les enregistrements viennent d'être décalés pour permettre d'insérer le nouveau titre.

```
1160 FOR K=1 TO 3
1170 A$(I,K)=S$(K)
1180 NEXT K
```

Insertion des 3 zones du nouveau titre et retour en début de séquence.

```
1190 GOTO 860
1200 NEXT I
1210 GOTO 860
```

Il nous reste encore quelques séquences à voir la semaine prochaine, en particulier la sauvegarde du fichier, la sortie sur imprimante et la liste à l'écran.

Tout cela va t'aider à terminer ta propre gestion de fichier pour nous l'envoyer le plus vite possible. QUI VA ETRE LE MEILLEUR dans chaque standard?? Les semaines qui viennent vont être chaudes. Rappelle-toi pour gagner il faut écrire uniquement en Basic et le plus court possible. Good Luck!!!

Bien Joystiquement Vôtre.
François LE GRIGUER

LA SEMAINE PROCHAINE
La suite de la gestion de fichier de Guy PARRENIN.



PAR FRANÇOIS LE GRIGUER



**SALUT LES PETITS GENIES!!
J'ESPERE QUE VOTRE
GESTION DE FICHER AVANCE
POUR POUVOIR, TRES BIEN-
TOT, LES EXPLOITER DANS
NOTRE INITIATION.
LA QUESTION EST
TOUJOURS LA MEME:
QUI SERA LE MEILLEUR??
ET CELA DANS CHAQUE
STANDARD. NOUS SAVONS
TOUS QUE LES LECTEURS DE
JOYSTICK Hebdo SE
DISTINGUENT EN
PROGRAMMATION ET,
NOUS ATTENDONS AVEC
IMPATIENCE DE RECEVOIR
VOS REALISATIONS POUR LE
PROUVER SUR
AMSTRAD, ST, AMIGA, PC,
C64, THOMSON.....**

J'APPREND

JOYSTICK FICHER

LE MEILLEUR SUR AMSTRAD CPC

Actuellement Guy PARRENIN reste le meilleur sur AMSTRAD mais peut-être pas pour très longtemps d'autres listings arrivent. En attendant on continue à commenter son listing de gestion de fichier avec les séquences: liste à l'écran, modification et sauvegarde du fichier.

1220 '-----LISTE ECRAN-----'

Cette liste doit afficher, par ordre alphabétique le fichier en cours, jeux ou utilitaires, page par page, avec trois options en fin de page: retour au menu, suite ou reprise de l'affichage au début. Le nombre total de titres est donné en haut de page.

1230 CLS

1240 LOCATE 60,1:PRINT IV\$;- TOTAL : -;IV\$

1250 LOCATE 4,3:PRINT IV\$;- DK LOGICIEL-;

1260 PRINT SPACE\$(25);-OBSERVATIONS-;SPACE\$(20);IV\$

Les premières lignes d'instructions affichent le haut de la page avec l'en-tête des colonnes et le nom du fichier traité.

1270 WINDOW#3,6,75,4,24

Déclaration d'une fenêtre sur le canal 3 représentant la surface d'affichage des lignes du fichier.

1280 N1=0

N1 contient le nombre de lignes éditées et N2 le nombre total de lignes du fichier.

1290 CLS#3

Cette instruction efface la surface d'écran suivant les paramètres de la fenêtre du canal 3.

1300 FOR I=1 TO 19

Début de la boucle de programme prévue pour afficher 19 lignes à l'écran.

1310 N1=N1+1

1320 IF N1>N2 THEN 1380

Contrôle si N1>N2 le fichier est édité en totalité.

1330 LOCATE 5,4+I:PRINT- -;SPACE\$(3-LEN(A\$(N1,2)));

En LOCATE la valeur de Y est définie par l'addition de la valeur de I (1 à 19). L'affichage du code numérique qui peut être 1, 2 ou 3 caractères est précédé éventuellement d'espaces par SPACE\$. LEN donne le nombre de caractères. 3-LEN définit le nombre d'espaces. Cette valeur numérique est la 2ème dans l'enregistrement et N1 contient le numéro d'enregistrement (N1,2).

1340 PRINT A\$(N1,2);- -;A\$(N1,1);

Affichage de la zone 1 (N1,1) qui contient le titre.

1350 PRINT STRING\$(32-LEN(A\$(N1,1)),46);- -;A\$(N1,3)

Le nombre maxi de caractères du titre 32 moins le nombre réel donne le nombre d'espaces dans lesquels on affiche des points (valeur ASCII du point 46). Affichage de la zone 3 (N1,3).

1360 LOCATE 68,1:PRINT IV\$;N1;SPACE\$(1);IV\$

1370 NEXT

Fin de l'affichage des 19 lignes.

1380 T1\$=- S : Suite D : Debut R : Retour menu -

1390 LOCATE(80-LEN(T1\$))/2,25:PRINT IV\$;T1\$;IV\$

Affichage en bas d'écran des options possibles.

1400 Z\$=INKEY\$:IF Z\$=- THEN 1400

1410 IF Z\$=-R THEN 130

1420 IF Z\$=-S AND N1<N2 THEN 1290


1430 IF Z\$=-D THEN 1280

Contrôle de la touche utilisée.

1440 GOTO 1400

1450 '-----MODIFICATIONS-----'

Une séquence qui permet de reprendre un enregistrement du fichier



pour modifier soit le code numérique (zone 2) soit les observations (zone 3) ou encore pour le supprimer.

```
1460 CLS
1470 WINDOW#2,10,70,5,21
1480 T1$=" MODIFICATIONS LOGITHEQUE "
1490 LOCATE(80-LEN(T1$))/2,4:PRINT IV$;T1$;IV$
Affichage avec calcul des espaces pour le centrage du texte. L'inversion vidéo est donnée par la chaîne IV$.
```

```
1500 CLS#2
1510 LOCATE 15,23:PRINT IV$; ENTER : Retour
menu;SPACE$(30);IV$
1520 LOCATE 41,23:PRINT IV$;SPACE$(24);IV$
1530 LOCATE 35,13:PRINT CHR$(18)
1540 LOCATE 22,13:LINE INPUT-QUEL LOGICIEL ? : -,N9$
Saisie du titre à modifier. Le titre peut être composé partiellement si les premiers caractères sont différents des titres voisins.
1550 IF N9$="" THEN 130
Retour en menu par Return.
1560 FOR I=1 TO N2
Boucle pour la recherche dans le fichier du titre saisi du numéro 1 au numéro maxi.
```

```
1570 IF N9$=LEFT$(A$(I,1),LEN(N9$)) THEN 1630
Comparaison du titre en fichier avec le titre saisi. Si identique suite en 1630.
```

```
1580 NEXT
1590 LOCATE 45,23:PRINT IV$; LOGICIEL INCONNU -;IV$
1600 PRINT CHR$(7)
1610 IF INKEY$="" THEN 1610
1620 GOTO 1520
```

Le titre n'existe pas: message, bip sonore et retour sur saisie.

```
1630 CLS#2
1640 CALL &BB03
1650 LOCATE 26,8:PRINT-1-NOM : -:A$(I,1)
1660 LOCATE 26,10:PRINT-2-DISC : -:A$(I,2)
1670 LOCATE 26,12:PRINT-3-OBS. : -:A$(I,3)
Affichage des trois lignes avec le libellé et le contenu de l'enregistrement en fichier.
```

```
1680 T1$=" M:Modif de 2 ou 3 S:Suppression R:Retour "
1690 LOCATE(80-LEN(T1$))/2,23:PRINT IV$;T1$;IV$
1700 Z$=INKEY$:IF Z$="" THEN 1700
1710 IF Z$="R" THEN 1500
1720 IF Z$="S" THEN 1750
1730 IF Z$="M" THEN 1860
```

Contrôle de la touche R, S ou M.

```
1740 GOTO 1700
```

Instructions pour la suppression d'un enregistrement.

```
1750 IF I=N2 THEN 1810
```

Traitement particulier s'il s'agit du dernier titre du fichier.

```
1760 FOR J=I TO N2-1
```

Boucle de programme de la position en cours dans le fichier jusqu'à la fin -1, pour assurer le décalage vers le haut de tous les enregistrements qui suivent celui qui vient d'être supprimé.

```
1770 FOR K=1 TO 3
```

```
1780 A$(J,K)=A$(J+1,K)
```

Boucle de transfert des 3 zones de la position fichier +1 (J+1,K) vers la position en cours (J,K).

```
1790 NEXT
1800 NEXT
```

```
1810 FOR K=1 TO 3
```

```
1820 A$(N2,K)=""
```

La dernière position du fichier est vidée ("").

```
1830 NEXT
```

```
1840 N2=N2-1
```

Le nombre total d'enregistrements est diminué de 1.

```
1850 GOTO 1500
```

Instructions pour la modification de la zone 2 ou 3.

```
1860 LOCATE 26,16:PRINT-QUELLE ZONE ? (2 ou 3) :-
```

```
1870 Z$=INKEY$:IF Z$="" THEN 1870
```

```
1880 IF Z$="2" OR Z$="3" THEN 1890 ELSE 1870
```

```
1890 Z=VAL(Z$)
```

```
1900 LOCATE 10,19:PRINT-ENTREZ LA NOUVELLE DONNEE :-
```

```
1910 LOCATE 38,19:IF Z=2 THEN PRINT-...-
```

```
1920 IF Z=3 THEN PRINT STRING$(32,46)
```

```
1930 LOCATE 38,19:LINE INPUT-,,Z$
```

```
1940 IF (Z$="" OR LEFT$(Z$,1)="" AND Z=1) THEN 1500
```

```
1950 IF Z<>1 THEN A$(I,Z)=Z$
```

```
1960 GOTO 1500
```

```
1970 '-----SAUVEGARDE-----
```

Écriture du fichier, modifié ou nouvellement créé, sur la disquette. Avec une cassette tu dois prévoir une cassette par fichier, sinon les problèmes de reboisement et de compteur sont assez difficiles à résoudre. Une autre cassette doit contenir le programme.

```
1980 CLS
```

```
1990 T1$=" SAUVEGARDE "
```

```
2000 LOCATE(80-LEN(T1$))/2,6:PRINT IV$;T1$;IV$
```

```
2010 T2$=" :T3$=" :T4$="
```

```
2020 T2$=" DE VOTRE LOGITHEQUE "
```

```
2030 LOCATE(80-LEN(T2$))/2,8:PRINT IV$;T2$;IV$
```

```
2040 T3$=" JEUX OU UTILITAIRES ? (J/U) "
```

```
2050 LOCATE(80-LEN(T3$))/2,13:PRINT T3$
```

```
2060 LOCATE 14,14:PRINT STRING$(7,154);SPACE$(8);
```

```
2070 PRINT STRING$(21,154);SPACE$(9);STRING$(5,154)
```

```
2080 Z$=INKEY$:IF Z$="" THEN 2110
```

```
2090 IF Z$="J" THEN 2120
```

```
2100 IF Z$="U" THEN 2250
```

Affichage du titre et choix. Contrôle de la touche U ou J.

```
2110 GOTO 2080
```

```
2120 '...(Sauvegarde du fichier JEUX)...
```

```
2130 T4$=" J E U X "
```

```
2140 LOCATE(80-LEN(T4$))/2,19:PRINT IV$;T4$;IV$
```

```
2150 aa$="JEUX.LOG";|ERA,@aa$
```

Avant l'écriture du fichier, l'ancien est effacé.

```
2160 OPENOUT "JEUX.LOG"
```

```
2170 WRITE#9,N2
```

Ouvrir le fichier JEUX.LOG et écrire N2 c-à-d le nombre total d'enregistrements.

```
2180 FOR I=1 TO N2
```

Début de la boucle d'écriture des enregistrements.

```
2190 FOR J=1 TO 3
```

Début de la boucle pour les 3 zones de chaque enregistrement.

```
2200 WRITE#9,A$(I,J)
```

```
2210 NEXT
```

```
2220 NEXT
```

```
2230 CLOSEOUT
```

Fermer le fichier

```
2240 PRINT CHR$(7):GOTO 130
```

Bip sonore et retour en menu.

```
2250 '...(Sauvegarde du fichier UTILITAIRES)...
```

La séquence est identique à la précédente.

```
2260 T4$=" UTILITAIRES "
```

```
2270 LOCATE(80-LEN(T4$))/2,19:PRINT IV$;T4$;IV$
```

```
2280 aa$="UTILIT.LOG";|ERA,@aa$
```

```
2290 OPENOUT "UTILIT.LOG"
```

```
2300 WRITE#9,N2
```

```
2310 FOR I=1 TO N2
```

```
2320 FOR J=1 TO 3
```

```
2330 WRITE#9,A$(I,J)
```

```
2340 NEXT
```

```
2350 NEXT
```

```
2360 CLOSEOUT
```

```
2370 PRINT CHR$(7):GOTO 130
```

**Satisfaits ?
Des suggestions ?
Écrivez à
François Le Griguer**

Pas mal la gestion de fichier de Guy PARRENIN, mais tu peux faire plus court. Par exemple, les répétitions de séquence comme la sauvegarde spécifique pour chaque fichier doit se regrouper et permettre de gérer un grand nombre de fichiers. A toi de faire mieux, moins de lignes programme et, pourquoi pas, plus de performances. Et bien sûr, toujours en Basic, l'Assembleur n'apporte pas grand-chose dans un traitement de ce genre. Dès que tu es prêt, tu m'envoies ton programme, je compte sur toi.

**Bien Joystiquement Vôtre.
François LE GRIGUER**

**LA SEMAINE PROCHAINE
La sortie sur imprimante des fichiers**



LES LISTINGS DE GESTION DE FICHIERS ARRIVENT EN GRAND NOMBRE, MAIS SEULEMENT SUR AMSTRAD CPC. WAOUH!! CA MARCHE TRES FORT!! VOUS ETES FORMIDABLES!! MAIS LES AUTRES, TOUS LES AUTRES, QUE SE P ASSE-T-IL?? CE QUE NOUS RECEVONS NE PEUT SE PUBLIER, LE NIVEAU RESTE VRAIMENT FAIBLE. LA GESTION DE FICHER NE VOUS INSPIRE P AS?? N'Y A-T-IL QUE LES AMSTRADISTES A SAVOIR PROGRAMMER?? QUE FONT LES ST, PC, C64...?? VOUS DEVEZ, VOUS AUSSI, AVOIR DES IDEES, NON?? IL FAUT RELEVER LE DEFI, NOUS VOUS ATTENDONS A VEC IMPATIENCE.



PAR FRANÇOIS LE GRIGUER

JOYSTICK FICHER

LA SORTIE DES FICHIERS SUR IMPRIMANTE

Suite et fin pour le listing de Guy PARRENIN sur AMSTRAD, avec la sortie sur imprimante, un programme de gestion de fichier qui a débuté dans notre numéro 20. Pour toi qui prends la rubrique en marche je rappelle qu'une grande compétition est ouverte entre tous les lecteurs de JOYSTICK Hebdo, les meilleurs et les plus "bos"!! pour la réalisation d'une gestion de fichiers, et ceci quel que soit ton micro. Il s'agit de faire plus fort que le voisin, en Basic, le plus court et le plus performant possible. Les listings sont publiés ici, en y ajoutant toutes les explications en italique pour ceux qui veulent apprendre. Tu charges ton programme par LOAD"JOYFICH" et tu saisis les lignes qui suivent, ensuite tu sauvegardes par SAVE"JOYFICH".

```

2380 '-----IMPRESSION-----
2390 CLS
2400 LOCATE 16,4:PRINT IV$;"
IMPRESSION";SPACES(20);"LOGITHEQUE ";IV$
2410 WINDOW#4,10,75,6,19
2420 CLS#4
Déclaration d'une fenêtre par WINDOW sur le canal 4 et effacement.
2430 LOCATE 16,20:PRINT IV$;" ";SPACES(14);
2440 PRINT"ENTER : Retour menu";SPACES(16);IV$
2450 D0$=""
2460 LOCATE 43,7:PRINT CHR$(18)
2470 LOCATE 24,7:LINE INPUT"DATE D'IMPRESSION : ";D0$
Saisie de la date du jour dans le format: xx.xx.xx
2480 IF D0$="" THEN 130
2490 D$=LEFT$(D0$,2):D=VAL(D$)
2500 IF D>31 THEN PRINT CHR$(7):GOTO 2450
Extraction du jour, 2 premiers caractères par LEFT$(D0$,2), et controle
si > 31.
2510 A$=MID$(D0$,4,2):A=VAL(A$)
2520 IF A>12 THEN PRINT CHR$(7):GOTO 2450
Extraction du mois, 2 caractères à partir du 4ème par MID$(D0$,4,2).
Controle mois >12.
2530 T$=RIGHT$(D0$,2):T=VAL(T$)
2540 IF T<>89 THEN PRINT CHR$(7):GOTO 2450
Extraction de l'année, 2 caractères à droite par RIGHT$(D0$,2), et
controle si différent de 89 (<>).
2550 IF A=1 THEN AA$="JANVIER"
2560 IF A=2 THEN AA$="FEVRIER"
2570 IF A=3 THEN AA$="MARS"
2580 IF A=4 THEN AA$="AVRIL"
2590 IF A=5 THEN AA$="MAI"
2600 IF A=6 THEN AA$="JUIN"
2610 IF A=7 THEN AA$="JUILLET"
2620 IF A=8 THEN AA$="AOUT"
2630 IF A=9 THEN AA$="SEPTEMBRE"
2640 IF A=10 THEN AA$="OCTOBRE"
2650 IF A=11 THEN AA$="NOVEMBRE"
2660 AA$="DECEMBRE"

```

Toutes les lignes qui précèdent servent à définir le contenu de AA\$ en fonction de la valeur de A (1 à 12). Et voici comment faire plus court:

- en Data mettre les 12 libellés des mois
- par une petite boucle mettre à jour un fichier MOISS
- ensuite AA\$ prend directement la valeur du fichier MOISS(A)

```

DATA "JANVIER","FEVRIER","MARS","AVRIL" etc....
FOR J=1 TO 12:READ MOISS(J):NEXT
AA$=MOISS(A)


```

```

2670 D0$=D$+" "+AA$+" "+T$
Formatage de la date avec les 2 caractères du jour, suivis du mois en
alpha et des 2 caractères de l'année.
2680 LOCATE 44,7:PRINT D0$
2690 LOCATE 30,20:PRINT IV$;SPACES(20);IV$
2700 LOCATE 23,10:PRINT"LOGICIEL A IMPRIMER ( J/U )"

```

J'APPRENDS



**Satisfaits ?
Des suggestions ?
Ecrivez à
François Le Griguer**

```

2710 T2$=""
2720 T2$=INKEY$:IF T2$="" THEN 2720
2730 IF T2$="J" THEN LOCATE 34,13:PRINT IV$;"J E U X";IV$:
      GOTO 2760
2740 IF T2$<>"U" THEN 2710
2750 LOCATE 28,13:PRINT IV$;" UTILITAIRES ";IV$
2760 LOCATE(80-LEN(Q$))/2,16:PRINT "(";Q$;")"
2770 Z$=INKEY$:IF Z$="" THEN 2770
2780 IF Z$="N" THEN 2420 ELSE IF Z$<>"O" THEN 2770
2790 CLS#4
2800 T1$="PREPAREZ VOTRE IMPRIMANTE ET PRESSEZ UNE TOUCHE"
2810 LOCATE(80-LEN(T1$))/2,11:PRINT T1$
2820 LOCATE 17,12:PRINT STRING$(47,154):PRINT CHR$(7)
2830 IF INKEY$="" THEN 2830 ELSE CLS#4
2840 T1$="IMPRESSION EN COURS"
2850 LOCATE(80-LEN(T1$))/2,11:PRINT T1$
2860 LOCATE 31,12:PRINT STRING$(19,154):PRINT CHR$(7)
Toutes les lignes précédentes ne posent aucun problème particulier,
elles renseignent l'utilisateur sur la suite du traitement.
2870 N3=0
2880 N4=0
2890 GOSUB 3150
Suite dans le sous-programme d'en-tête de la page.
2900 FOR I=1 TO N2
Début de la boucle d'édition du fichier (N2 contient le nombre total).
L'édition sur l'imprimante se différencie d'un affichage écran seulement
au niveau de l'instruction PRINT qui est suivie du numéro de canal 8
(PRINT#8).
2910 IF N1<55 THEN 2960
Contrôle du nombre de lignes par page pour l'imprimante.
2920 PRINT#8,"I      I";SPACE$(34);"I";SPACE$(35);"I"
2930 PRINT#8," ";STRING$(78,45)
2940 PRINT#8:PRINT#8:PRINT#8:PRINT#8:PRINT#8
Fin d'une page sur l'imprimante.
2950 GOSUB 3150
2960 PRINT#8,"I      ";SPACE$(3-LEN(A$(I,2)));
2970 PRINT#8,A$(I,2);" I ";A$(I,1);STRING$(32-LEN(A$(I,1)),46);
2980 PRINT#8," I ";A$(I,3);SPACE$(34-LEN(A$(I,3)));"I"
Edition de la ligne sur l'imprimante avec le calcul des espaces comme
dans l'affichage écran.
2990 N1=N1+1
3000 N4=N4+1
3010 NEXT I
Fin de la boucle d'édition.
3020 PRINT#8,"I      I";SPACE$(34);"I";SPACE$(35);"I"
3030 PRINT#8," ";STRING$(78,45)
3040 PRINT#8
3050 PRINT#8," TOTAL :";N4
3060 CLS#4
3070 T1$="UNE AUTRE IMPRESSION ? (O/N)"
Fin de l'édition et choix.
3080 LOCATE(80-LEN(T1$))/2,11:PRINT T1$
3090 LOCATE 23,12:PRINT STRING$(34,154)
3100 LOCATE 30,20:PRINT IV$;SPACE$(20);IV$;CHR$(7)
3110 Z$=INKEY$:IF Z$="" THEN 3140
3120 IF Z$="O" THEN 2790
3130 IF Z$="N" THEN 130
3140 GOTO 3110
sous-programme en-tête listing
3150 L$=""

```

```

3160 IF T2$="J" THEN L$="JEUX"
3170 IF T2$="U" THEN L$="UTILITAIRES"
3180 N3=N3+1
3190 N1=0
3200 PRINT#8
3210 PRINT#8
3220 IF T2$="J" THEN PRINT#8,CHR$(18);CHR$(14);
      SPACES$(22-LEN(L$));L$
3230 IF T2$="U" THEN PRINT#8,CHR$(18);CHR$(14);
      SPACES$(25-LEN(L$));L$
3240 PRINT#8
3250 PRINT#8," ";D0$;SPACES$(67-LEN(D0$));"PAGE";N3
3260 PRINT#8
3270 PRINT#8," ";STRING$(78,45)
3280 PRINT#8,"I DISK I";SPACE$(12);"LOGICIEL";SPACE$(14);
3290 PRINT#8,"I";SPACE$(11);"OBSERVATIONS";SPACE$(12);"I"
3300 PRINT#8,"I";STRING$(78,45);"I"
3310 PRINT#8,"I      I";SPACE$(34);"I";SPACE$(35);"I"
3320 RETURN
3330 '-----EFFACEMENT DU FICHER EN MEMOIRE---
Une séquence toute simple pour l'effacement de la totalité d'un fichier
en mémoire.
3340 CLS
3350 T1$=" EFFACEMENT "
3360 LOCATE(80-LEN(T1$))/2,9:PRINT IV$;T1$;IV$
3370 T2$=" DU FICHER EN MEMOIRE "
3380 LOCATE(80-LEN(T2$))/2,11:PRINT IV$;T2$;IV$
3390 LOCATE(80-LEN(Q$))/2,15:PRINT Q$
3400 LOCATE 36,16:PRINT STRING$(8,154)
3410 PRINT CHR$(7)
3420 Z$=INKEY$:IF Z$="" THEN 3450
3430 IF Z$="O" THEN 3460
3440 IF Z$="N" THEN 130
3450 GOTO 3420
3460 CLEAR
3470 FOR T=1 TO 200:NEXT T:PRINT CHR$(7)
3480 GOTO 110
3490 '-----RETOUR BASIC-----
Le retour au Basic efface l'ensemble des programmes et fichiers en
mémoire.
3500 CALL 0

```

La gestion de fichiers, détaillée et commentée, de Guy PARRENIN s'achève avec cette séquence. Un très bon exercice en basic, c'était aussi l'occasion de découvrir une autre catégorie de logiciels. Vous êtes supers, de nombreuses "gestion de fichier" viennent de nous parvenir. La semaine prochaine je vous propose un changement d'air, un autre style mais surprise... surprise...

*Bien Joystiquement Vôtre.
François LE GRIGUER*

LA SEMAINE PROCHAINE
Le début d'une toute nouvelle série à ne manquer sous aucun prétexte !!

ABONNEZ-VOUS LES AVANTAGES :

- 40 F d'économie

Un abonnement pour 6 mois (24 N°s vous reviennent à 200 F au lieu de 240 F, soit l'équivalent de 4 N°s gratuits)

- Toutes vos P.A. gratuites

- **125 F de réduction** sur l'achat d'un **Multiface** à commander directement chez POWER PRODUCT FRANCE (Pour ST - COMMODORE - CPC - SPECTRUM) et qui vous permet, entre autres, la sauvegarde de vos parties en cours et de pouvoir placer vos POKES de vies infinies.

BULLETIN D'ABONNEMENT

à retourner sous enveloppe affranchie avec le règlement par chèque ou CCP à **JOYSTICK Hebdo** Service abonnement, 177, rue St-Honoré - 75001 PARIS

OUI, je m'abonne pour 6 mois (24 n°) à 200 F au lieu de 240 F.

NOM
 PRENOM
 ADRESSE
 CODE POSTAL VILLE
 AGE ORDINATEUR SIGNATURE